Problem:
Score : 300 points

Problem Statement

We have two permutations P and Q of size N (that is, P and Q are both rearrangements of (1,~2,~...,~N)).

There are N! possible permutations of size N. Among them, let P and Q be the a-th and b-th lexicographically smallest permutations, respectively. Find |a - b|.

Notes

For two sequences X and Y, X is said to be lexicographically smaller than Y if and only if there exists an integer k such that X_i = Y_i~(1 \leq i < k) and X_k < Y_k.

Constraints

2 \leq N \leq 8

P and Q are permutations of size N.

Input

Input is given from Standard Input in the following format:

N
P_1 P_2 ... P_N
Q_1 Q_2 ... Q_N

Output

Print |a - b|.

Sample Input 1

3
1 3 2
3 1 2

Sample Output 1

3

There are 6 permutations of size 3: (1,~2,~3), (1,~3,~2), (2,~1,~3), (2,~3,~1), (3,~1,~2), and (3,~2,~1). Among them, (1,~3,~2) and (3,~1,~2) come 2-nd and 5-th in lexicographical order, so the answer is |2 - 5| = 3.

Sample Input 2

8
7 3 5 4 2 1 6 8
3 8 2 5 4 6 7 1

Sample Output 2

17517

Sample Input 3

3
1 2 3
1 2 3

Sample Output 3

0

Solution in Pascal (FPC 3.0.4) (judge verdict: Accepted)

program main;
const
  size = 8;

type
  sample = array [1..size] of longint;

var
  n, order : longint;
  p, q : sample;


  procedure input (var a : sample);
  var
    i : longint;

    begin
      for i := 1 to n do
        read(a[i]);
      readln;
    end;

  function code (var a : sample) : sample;
  var 
    i, j, c : longint;
  
    begin
      for i := 1 to n do
        a[i] := a[i] - 1;  //zero index
      for i := 1 to n do begin
        c := 0;
        for j := 1 to (i-1) do
          if (a[j] < a[i]) then inc(c);
        code[i] := a[i] - c;
      end;  //lehmer code
    end;

  function rank (a : sample) : longint;
  var
    i, t : longint;	

      function fact (n : longint) : longint;
        begin
          if (n = 0) or (n = 1) then fact := 1
          else fact := n * fact(n-1);
        end;
  
    begin
      t := 0;
      for i := 1 to n do begin
        t := t + a[i]*fact(n-i);
      end;  //lexicographic rank
	  inc(t);
	  rank := t;
    end;


begin
  readln(n);
  input(p);
  input(q);
  order := abs(rank(code(p))-rank(code(q)));
  writeln(order);
end.